log using MattesWeeksIO-Reacting, replace

* Replication file for Mattes and Weeks "REACTING TO THE OLIVE BRANCH: HAWKS, DOVES, AND PUBLIC SUPPORT FOR COOPERATION"
version 16

clear

* Study A (U.S. strong; fielded in May 2021)
clear
import delimited "MattesWeeksIO-StudyA.csv", bindquote(strict)
gen studya=1
gen studyb=0
save studyA.dta, replace
clear

* Study B (U.S. weak; fielded in April 2021)
clear
import delimited "MattesWeeksIO-StudyB.csv", bindquote(strict)
gen studya=0
gen studyb=1
gen parity=. 
save studyB.dta, replace

use studyA.dta, clear
append using studyB.dta, force

****************************************
* CLEAN UP DATA AND GENERATE VARIABLES *
****************************************

keep if status==0 // previews
count // 

*********************************
* PRETREATMENT ATTENTION CHECKS *
*********************************

* pretreatment attention checks
foreach x in 1 2 3 {
gen attn`x'p=0
replace attn`x'p=1 if attn`x'=="pass"
}
gen passpresum=attn1p + attn2p + attn3p
tab passpresum

gen passpre1=0
replace passpre1=1 if attn1=="pass" & attn2=="pass" & attn3=="pass"
tab passpre1 if studya==1 // this shows what % of subjects did not pass the pretreatment attention checks: they did not proceed to experiment
tab passpre1 if studyb==1 // this shows what % of subjects did not pass the pretreatment attention checks: they did not proceed to experiment

drop if passpre1==0
count

count if studya==1
count if studyb==1

**************
* TREATMENTS *
**************

*whether the U.S. president is a Democrat 
gen dem=0
replace dem=1 if party=="Democratic Party"
tab dem party, nomiss

*whether the U.S. president is a Republican
gen rep=0
replace rep=1 if party=="Republican Party"
tab rep party, nomiss

**whether the U.S. president is a hawk
gen hawk_us=0
replace hawk_us=1 if hawk_t=="hawk" // US hawk
tab hawk_us hawk_t, nomiss

*whether the U.S. president is a dove
gen dove_us=0
replace dove_us=1 if hawk_t=="dove" // US dove
tab dove_us hawk_t, nomiss

*whether the foreign leader is a hawk
gen hawk_f=0
replace hawk_f=1 if hawkf_t=="hawk" // Russia hawk
tab hawk_f hawkf_t, nomiss

*whether the foreign leader is a dove 
gen dove_f=0
replace dove_f=1 if hawkf_t=="dove" // Russia dove
tab dove_f hawkf_t, nomiss
label define dovel 0 "Foreign Hawk" 1 "Foreign Dove"
label values dove_f dovel

*whether the gesture was low-cost ("cheap) or costly ("strong')
gen rapproche=1 if rapf_t=="cheap" // gesture strength
replace rapproche=2 if rapf_t=="strong"

label define rapprochel 1 "Cheap" 2 "Strong"
label values rapproche rapprochel
tab rapproche rapf_t, nomiss

* Interactions between foreign leader reputation and gesture strength
gen hawkhi=0
replace hawkhi=1 if hawk_f==1 & rapproche==2
tab rapproche hawkhi

gen dovehi=0
replace dovehi=1 if hawk_f==0 & rapproche==2

gen hawklo=0
replace hawklo=1 if hawk_f==1 & rapproche==1

gen dovelo=0
replace dovelo=1 if hawk_f==0 & rapproche==1


**************
* COVARIATES *
**************

** Age

tab age // lucid age
cor age agemw // Mattes-Weeks age


** Party

*7-point 
gen pid7 = .
replace pid7 = 1 if pid_1 == 2 & pid_2 == 1  // strong democrat
replace pid7 = 2 if pid_1 == 2 & pid_2 == 2  // weak democrat
replace pid7 = 3 if pid_3 == 2  // lean democrat
replace pid7 = 4 if pid_3 == 3  // doesn't lean
replace pid7 = 5 if pid_3 == 1  // lean republican
replace pid7 = 6 if pid_1 == 1 & pid_2 == 2  // weak republican
replace pid7 = 7 if pid_1 == 1 & pid_2 == 1  // strong republican
label define pid7l 1 "Strong Democrat" 2 "Weak Democrat" 3 "Lean Democrat" 4 "Don't Lean" 5 "Lean Republican" 6 "Weak Republican" 7 "Strong Republican"
label values pid7 pid7l
tab pid7

* 5-point 
gen pid5 = .
replace pid5=pid_1
label define pid5l 1 "Republican" 2 "Democrat" 3 "Independent" 4 "Another party" 5 "No preference"
label values pid5 pid5l
tab pid5

* 3-point with leaners coded as "other"
* use pid5
gen pid3=0
replace pid3=1 if pid5==2
replace pid3=2 if pid5==1
replace pid3=3 if pid5==3 | pid5==4 | pid5==5
label define pid3l 1 "Democrat (no leaners)" 2 "Republican (no leaners)" 3 "Other"
label values pid3 pid3l

* pid3 by study
tab pid3 if studya==1
tab pid3 if studyb==1

gen dem3=0
replace dem3=1 if pid3==1 // no leaners

gen rep3=0
replace rep3=1 if pid3==2 // no leaners

gen ind3=0
replace ind3=1 if pid3==3 // everyone else

* rescale pid7 to have min 0 max 1
drop pid_*
rename pid7 pid7_f
gen pid7=(pid7_f-1)/6

* nonpartisan
gen nonpartisan=0
replace nonpartisan=1 if pid5==3 | pid5==5


** Political Ideology

* ideologically moderate
tab ideo7
gen moderate=0
replace moderate=1 if ideo7==3 | ideo7==4 | ideo7==5

* left vs. right
* ideo7  1 = extremely liberal and 7 = extremely conservative
gen left=0
replace left=1 if ideo7<4
tab left
gen right=0
replace right=1 if ideo7>4

replace ideo7=(ideo7-1)/6
tab ideo7

* ideology // lower values mean more liberal, higher values more conservative (MM MOVED THIS HERE FROM BELOW)
gen conservative=ideo7


** Gender

* male
gen male = cond(gender == 1,1,0) 
label define malel 1 "Male" 0 "Other"
label values male malel
tab male


** Hawkishness

*higher values = more hawkish
/*
hawk7 Now we are going to ask you a few questions about your personal views.Some people think that U.S. military force should never be used under any circumstances. They are at “1” on the scale below. Other people think there are many situations in which U.S. military force should be used to deal with problems. They are at “7” on the scale below. And, of course, other people have opinions in between.Where would you put yourself on this scale?
?	1 -- U.S. military force should never be used under any circumstances (1)
?	2 (2)
?	3 (3)
?	4 (4)
?	5 (5)
?	6 (6)
?	7 -- U.S. military force should be used in many situations (7)

hawk5 Please tell us how much you agree or disagree with the following statements:“The use of military force only makes problems worse.”
?	Agree strongly (1)
?	Agree somewhat (2)
?	Neither agree nor disagree (3)
?	Disagree somewhat (4)
?	Disagree strongly (5)
*/

rename hawk5 hawk5_f
gen hawk5= (hawk5_f-1)/4
rename hawk7 hawk7_f
gen hawk7= (hawk7_f-1)/6
tab hawk7

gen hawkind_f=hawk5_f+hawk7_f-2 // hawk index
gen hawkind=hawkind_f/10
tab hawkind
sum hawkind, d

* binary hawkishness variable
gen hawkish=0
replace hawkish=1 if hawkind>=.6 // above 50th percentile for both samples
tab hawkish


** Internationalism

/*
intl “The United States needs to play an active role in solving conflicts around the world.” 
?	Agree strongly (1)
?	Agree somewhat (2)
?	Neither agree nor disagree (3)
?	Disagree somewhat (4)
?	Disagree strongly (5)
*/
recode intl 1=5 2=4 3=3 4=2 5=1 // now higher values mean more internationalist

rename intl intl_f
gen intl= (intl_f-1)/4


** Trust

/*
* "Generally speaking, would you say that the U.S. can trust other nations, or that the U.S. should be very careful in dealing with other nations?"
* o	The U.S. can trust other nations  (1) 
* o	The U.S. should be very careful in dealing with other nations  (2) 
*/

tab trust
recode trust (2=0)
label define trustl 1 "Can trust" 0 "Should be very careful"
label values trust trustl
tab trust


** Education

* educ // higher = more education
label define educationmwl 1 "No High School" 2 "High School Grad" 3 "Some college" 4 "2-Year College Degree" 5 "4-Year College Degree" 6 "Postgraduate Degree"
label values educationmw educationmwl

/*
* Lucid education
tab educ
replace educ=. if educ<0
label define educll 1 "Some high school or less" 2 "High School Grad" 3 "Other post high school vocational" 4 "Some college" 5 "Associate's degree" 6 "Bachelor's degree" 7 "Master's or professional degree" 8 "PhD"
label values education educll
*/

* note: we use mw educ variable
gen college=0
replace college=1 if educationmw>=5 // college degree or no (doesn't count those in college)

gen nocollege=0
replace nocollege=1 if educationmw<5 // includes those w/ some college

rename educationmw educ_f
gen educ= (educ_f-1)/5
tab educ
tab educ_f college


** Race

* Lucid race variable
label define racel 1 "White" 2 "Black, or African American" 3 "American Indian or Alaska Native" 4 "Asian *** Asian Indian" 5 "Asian *** Chinese"  6 "Asian *** Filipino"  7 "Asian *** Japanese"  8 "Asian *** Korean" 9 "Asian *** Vietnamese" 10 "Asian *** Other" 11 "Pacific Islander *** Native Hawaiian"  12 "Pacific Islander *** Guamanian" 13 "Pacific Islander *** Samoan " 14 "Pacific Islander *** Other Pacific Islander" 15 "Some other race" 16	"Prefer not to answer" 
label values ethnicity racel

* white
gen white = cond(ethnicity == 1,1,0)
tab white


* Political Interest

/* Some people seem to follow what's going on in government and public affairs most of the time, whether there's an election going on or not. 
Others aren't that interested.    
 	Would you say you follow what's going on in government and public affairs...?
o	Most of the time  (1) 
o	Some of the time  (2) 
o	Only now and then  (3) 
o	Hardly at all  (4) 
o	Don't know  (5) 
*/

gen polint = newsint
recode polint 4=0 3=1 2=2 1=3 5=. *=. // recode so that higher values mean more interested [5 originally meant "don't know"]
									  // so now 0 = hardly at all, 1 = Only now and then, 2 = Some of the time, 3=most of the time
recode voted20 4=1 1/3=0 *=. // yes or not yes
recode voted18 4=1 1/3=0 *=. // yes or not yes

rename polint polint_f
label define polint_fl 0 "Hardly at all" 1 "Only now and then" 2 "Some of the time" 3 "Most of the time"
label values polint_f polint_fl
gen polint= polint_f/3
tab polint polint_f

*politically interested
gen polinthi=0
replace polinthi=1 if polint>.5
tab polinthi polint_f

tab particip
rename particip particip_string
foreach x in  1 2 3 4 5 {
gen particip_`x'=0
replace particip_`x'=1 if strpos(particip_s, "`x'")
}

egen particip_f = rowtotal(particip_1 particip_2 particip_3 particip_4) // ignore donate blood

gen particip = particip_f/4
tab particip


** Religious

recode relig_impt 4=1 3=2 2=3 1=4 *=. // higher values means more important
recode relig_born 1=1 2=0 *=. // dummy var

rename relig_impt relig_impt_f
gen relig_impt= (relig_impt_f - 1)/3


** Perception of Russia as an Enemy

/*enemy Please say whether you consider Russia an ally of the United States, friendly, but not an ally, unfriendly, or an enemy of the United States.
o	Ally  (1) 
o	Friendly, but not an ally  (2) 
o	Unfriendly  (3) 
o	Enemy  (4) 
o	No opinion  (5) */

gen enemy2 = 0
replace enemy2 = 1 if enemy== 3 | enemy==4 // what should we do with "no opinion"? Not an enemy? Alternatively could code as missing.

tab pid5, sum(enemy2) //  Republicans a bit less likely to think Russia is an enemy

* for paper:
tab pid3 if studya==1, sum(enemy2)
tab pid3 if studyb==1, sum(enemy2)


** Components for Elitelike Samples

* income
* lucid hhi variable
replace hhi=. if hhi==-3105
sum hhi, d
gen highinc=0
replace highinc=1 if hhi>8 // above the median in the sample
tab highinc

*"prime age"
gen primeage=0
replace primeage=1 if age>=40 & age<65
tab primeage

*high-educ
gen higheduc=0
replace higheduc=1 if educ>=.8
tab educ higheduc

* Interactions between elitelike indicators and dove
gen dovexmale=0 // interaction between foreign dove and male
replace dovexmale=1 if male==1 & dove_f==1
tab dovexmale
gen dovexinc=0 // interaction between foreign dove and high-income
replace dovexinc=1 if dove_f==1 & highinc==1 
tab dovexinc
gen dovexeduc=0 // interaction between foreign dove and education
replace dovexeduc=1 if dove_f==1 & educ>=.8
tab dovexeduc
gen dovexage=0 // interaction between foreign dove and "prime policymaker age"
replace dovexage=1 if dove_f==1 & age>=40 & age<65
tab dovexage
gen dovexpolint=0 // interaction between foreign dove and political interest
replace dovexpolint=1 if dove_f==1 & polint>.5
tab dovexpolint

* Generate different definitions of elites
gen elite1=0
replace elite1=1 if male==1 & highinc==1 & educ>.8 & primeage==1
label var elite1 "Male, high-inc, college+, age 40-64"
tab elite1

gen elite2=0
replace elite2=1 if highinc==1 & educ>.8 & primeage==1
label var elite2 "High-inc, college+, age 40-64"
tab elite2 

gen elite3=0
replace elite3=1 if highinc==1 & educ>.8 & primeage==1 & polinthi==1
label var elite3 "High-inc, college+, age 40-64, high political interest"
tab elite3

gen elite4=0
replace elite4=1 if educ>.8 & primeage==1 & polinthi==1
label var elite4 "College+, age 40-64, high political interest"
tab elite4

gen elite5=0
replace elite5=1 if educ>.8 & polinthi==1
label var elite5 "College+, high political interest"
tab elite5

gen elite6=0
replace elite6=1 if polinthi==1
label var elite6 "High political interest"
tab elite6


** DV AND MEDIATORS

* Main DV (Reciprocate)

/*
U.S. President Richards is considering whether to pursue a more cooperative or conflictual policy toward Russia. He is deciding what to do about U.S. bases in the Arctic.  Which of the following comes closest to your view? President Richards should...
o	Build additional U.S. bases in the Arctic  (0) 
o	Keep the same number of U.S. bases in the Arctic  (1) 
o	Reduce the number of U.S. bases in the Arctic  (2) 
*/
/* Our DV is a categorical variable with three possible outcomes: subjects will indicate whether they thought that the U.S. president should “reduce the number of U.S. bases in the Arctic” (2), “keep the same number of U.S. bases” (1), or “build additional U.S. bases” (0). For ease of interpretation, we will transform this DV into a binary variable: the DV “support for reciprocation” will be coded 1 for subjects who think the U.S. should reduce the number of bases and 0 for subjects who thought the U.S. should keep the same number of bases or increase their number. This DV can thus be interpreted as the percent of respondents that support the reciprocation of the foreign gesture.
*/

gen policy=dv1
label define policyl 0 "Build more" 1 "Keep same" 2 "Reduce" 
label values policy policyl
label values dv1 policyl
tab policy

gen recip=0
replace recip=100 if dv1==2


* Supplemental DV (Number of Bases)

* PAP: Our supplemental DV, NumberBases, is an interval measure ranging from 0 to 14 based on the number of bases the respondent says they would like the U.S. to have in the Arctic (0 if they support removing all of the bases, and 14 if they support adding 7 or more bases). Larger values indicate less willingness to reciprocate the foreign conciliatory gesture, i.e. greater U.S. assertiveness in the Arctic. 
* build In your opinion, how many additional U.S. military bases should President Richards build in the Arctic, beyond the 7 it already has?
* close In your opinion, how many of the 7 U.S. military bases should President Richards close in the Arctic?

tab dv1 build, miss
tab dv1 close, miss
gen numberbases=0

*If Study A (U.S. strong)
replace numberbases=9 if dv1==1 & studya==1
replace numberbases=9-close if dv1==2 & studya==1
replace numberbases=9+build if dv1==0 & studya==1
tab numberbases

*If Study B (U.S. weak)
replace numberbases=7 if dv1==1 & studyb==1
replace numberbases=7-close if dv1==2 & studyb==1
replace numberbases=7+build if dv1==0 & studyb==1
tab numberbases


* Sincerity Mediator (Main Analysis)

/*sincere1 Russian President ${e://Field/flast} wants peace with the U.S.
o	Strongly agree  (1) 
o	Somewhat agree  (2) 
o	Neither agree nor disagree  (3) 
o	Somewhat disagree  (4) 
o	Strongly disagree  (5) 
*/
gen sincere=0
replace sincere=100 if sincere1<=2


* Exploitation Mediator (Additional Hypothesis)

* First Exploitation variable: Tmeporarily Close"

/*
dv2 Do you agree or disagree with the following statement?    
dv2 Do you agree or disagree with the following statement?    
  The U.S. should temporarily close some bases in the Arctic, but re-open them and send additional forces once Russia has finished reducing its own forces in the Arctic.   
o	Agree strongly  (1) 
o	Agree somewhat  (2) 
o	Neither agree nor disagree  (3) 
o	Disagree somewhat  (4) 
o	Disagree strongly  (5) 
*/

gen exploit=0
replace exploit=100 if dv2<=2
tab exploit

* Second exploitation variable: "Trick"

/* exploitation The U.S. can trick Russian President ${e://Field/flast} into weakening Russia's position in the Arctic.  
o	Strongly agree  (1) 
o	Somewhat agree  (2) 
o	Neither agree nor disagree  (3) 
o	Somewhat disagree  (4) 
o	Strongly disagree  (5) 
*/

gen trick=0
replace trick=100 if exploitation<=2


* Fear Mediator (Additional Hypothesis)

/*
punish If the U.S. does not cooperate with Russia now, Russia will find a way to punish the U.S.  
o	Strongly agree  (1) 
o	Somewhat agree  (2) 
o	Neither agree nor disagree  (3) 
o	Somewhat disagree  (4) 
o	Strongly disagree  (5) 
*/
tab punish
gen fear=0
replace fear=100 if punish<=2


* Power Balance (Appendix A, fn. 5)

*Check for whether respondents care about power balance
/*“The U.S. must at all times maintain at least as many Arctic military bases as Russia.”
o	Strongly agree  (1) 
o	Somewhat agree  (2) 
o	Neither agree nor disagree  (3) 
o	Somewhat disagree  (4) 
o	Strongly disagree  (5) 
*/

tab parity
gen balance=0
replace balance=1 if parity<=2
tab balance



*************************
* POST-TREATMENT CHECKS *
*************************

*mck_party mck_hawk_us mck_hawk_rus mck_tense mck_scen2 mck_arc3 mck_arc2 mck_rap 

* manipulation checks

* US hawk/dove
tab hawk_t mck_hawk_us, nol
gen pmck_hawk_us=0
replace pmck_hawk_us=1 if mck_hawk_us==1
tab hawk_t pmck_hawk_us, row // 

* US party
tab party mck_party, row 
gen pmck_party=0
replace pmck_party=1 if (dem==1 & mck_party==2) | (dem==0 & mck_party==1)
tab party pmck_party, row // 

* Tense Relations
tab mck_tense
gen pmck_tense=0
replace pmck_tense=1 if mck_tense==1
tab pmck_tense

* Arctic 2
/*
mck_arc2 Please select the option that best describes the situation. In 2027....
o	Both the U.S. and Russia have a major military presence in the Arctic  (1) 
o	Russia has 9 military bases in the Arctic  (2) 
o	The U.S. has 7 military bases in the Arctic  (3) 
o	All of the above  (4) 
*/
tab mck_arc2, miss
gen pmck_arc2=0
replace pmck_arc2=1 if mck_arc2==4
tab pmck_arc2 // only 70% passed

* Arctic 3 // 
/*mck_arc3 Please select the option that best describes the situation. In 2027....
o	The U.S. is stronger than Russia overall, but Russia is stronger in the Arctic  (1) 
o	The U.S. and Russia played in a ping-pong tournament  (2) 
o	Russia gave the U.S. a commemorative plaque  (3) 
o	Russia held a folk-dancing competition  (4) 
*/
tab mck_arc3 
gen pmck_arc3=0
replace pmck_arc3=1 if mck_arc3==1
tab pmck_arc3

* Scen_2
/*
mck_scen2 Please select the option that best describes the situation. In 2027...
o	The Arctic is a key site of competition between the U.S. and Russia. U.S. officials have called the Arctic "the first line of defense."  (1) 
o	The U.S. and Russia will play against one another at the World Cup  (2) 
o	All trade between the U.S. and Russia will occur via air  (3) 
o	Russia had a major earthquake  (4) 
*/
tab mck_scen2
gen pmck_scen2=0
replace pmck_scen2=1 if mck_scen2==1
tab pmck_scen2

* Foreign hawk
/*
mck_hawk_rus Which of the following best describes President ${e://Field/flast}'s reputation?
o	He has a reputation for ${e://Field/hawkfs}  (1) 
o	He has a reputation for being a foodie  (2) 
o	He has a reputation for liking opera  (3) 
o	Not sure  (4) 
*/
tab mck_hawk_rus
gen pmck_hawk_rus=0
replace pmck_hawk_rus=1 if mck_hawk_rus==1
tab pmck_hawk_rus

* Rapproche
tab mck_rap rapproche
gen pmck_rap=0
replace pmck_rap=1 if mck_rap==1
tab pmck_rap

foreach var of varlist mck_par mck_hawk_us mck_hawk_rus mck_tense mck_scen2 mck_arc3 mck_arc2 mck_rap {
	tab p`var'
}


* How many failed post-treatment attentions checks

* mck_party mck_hawk_us mck_hawk_rus mck_tense mck_scen2 mck_arc3 mck_arc2 mck_rap 
gen passpostsum = pmck_party + pmck_hawk_us + pmck_hawk_rus + pmck_tense + pmck_scen2 + pmck_arc3 + pmck_arc2 + pmck_rap 
tab passpostsum // 
sum passpostsum, det
gen mck_rat=passpostsum/8
tab mck_rat

gen pass85=0
replace pass85=1 if mck_rat>=.85 // PER PAP
tab pass85 if studya==1 // Study A: 82%; Study B: only 78%. 
tab pass85 if studyb==1


**********************************
** CHECK FOR RANDOM ASSIGNMENT  **
**********************************

local cklist hawk5 hawk7 hawkind intl trust male age educ relig_impt ///
   relig_born ideo7 pid7 polint voted20 voted18 particip

* Study A - U.S.-Strong
preserve
keep if studya==1
count
   foreach v of varlist `cklist' {
   reg hawk_us dove_f, noheader
   reg dem dove_f, noheader
   di in r _n(2) "... Variable is `v' ..."
   reg `v' dove_f, noheader
   reg `v' hawk_us, noheader
   reg `v' rapproche, noheader
   reg `v' dem, noheader
   reg recip `v', noheader
   }

* Study B - U.S.-Weak
restore
preserve
keep if studyb==1
count
   foreach v of varlist `cklist' {
   reg hawk_us dove_f, noheader
   reg dem dove_f, noheader
   di in r _n(2) "... Variable is `v' ..."
   reg `v' dove_f, noheader
   reg `v' hawk_us, noheader
   reg `v' rapproche, noheader
   reg `v' dem, noheader
   reg recip `v', noheader
   }
 
restore
count



**************
** ANALYSIS **
**************

* define controls to use in each analysis
foreach var of varlist ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust {
tab `var'
}

local controls "ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust"

* descriptive stats
tab ideo7 // 1 = extremely liberal and 7 = extremely conservative
tab pid5
tab pid7
tab pid3
tab left
tab right
tab left right
tab male
tab white
tab trust
tab hawkind
tab pid7
sum age, d
tab educ
tab relig_impt 
tab polint 
tab voted20

* clean up some unneccessary variables
keep ideo7 hawk7_f trust enemy attn_1 ethn attn_2 hawk5_f intl_f attn_3 mck_party mck_hawk_us mck_hawk_rus mck_tense mck_scen2 mck_arc3 mck_arc2 mck_rap dv1 build close dv2 exploitation punish parity newsint voted18 voted20 particip_string educ_f racemw relig_impt_f relig_born agemw gendermw comments age gender hhi ethnicity hispanic education political_party region party-pass85

save MattesWeeksIO.dta, replace

* Run the analyses below using MattesWeeksIO.dta

*******************************************************************************************************************************************
* H1 (Dove’s Advantage): Individuals are more likely to support reciprocating a conciliatory gesture by a foreign dove than reciprocating *
* the same conciliatory gesture by a foreign hawk.			                        													  *
*******************************************************************************************************************************************

* FIGURE 1 (MANUSCRIPT)

reg recip i.dove_f if rapproche==1 & studya==1 
margins dove_f, post 
est store model1
reg recip i.dove_f if rapproche==1 & studyb==1 
margins dove_f, post 
est store model2

coefplot model1, bylabel("US-Strong Study") ///
|| model2, bylabel("US-Weak Study") /// what we're plotting
scheme(s1mono) /// scheme
mlabel format(%3.1f) mlabpos(12) byopts(compact cols(1)) ///
xtitle("                             Support for Reciprocation (%)") // 


* SUPPLEMENTAL ANALYSES (APPENDIX)

* Table C-1: Main Effect
* Study A
reg recip dove_f if rapproche==1 & studya==1
outreg2 using TableC-1.doc, replace ctitle (Model 1)
reg recip dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableC-1.doc, append ctitle (Model 2)
* Study B
reg recip dove_f if rapproche==1 & studyb==1
outreg2 using TableC-1.doc, append ctitle (Model 3)
reg recip dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableC-1.doc, append ctitle (Model 4)

* Table D-1: Main Effect - Number of Bases
* Study A
reg numberbases dove_f if rapproche==1 & studya==1
outreg2 using TableD-1.doc, replace ctitle (Model 1)
reg numberbases dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableD-1.doc, append ctitle (Model 2)
* Study B
reg numberbases dove_f if rapproche==1 & studyb==1
outreg2 using TableD-1.doc, append ctitle (Model 3)
reg numberbases dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableD-1.doc, append ctitle (Model 4)

* Table D-3: Main Effect - 85% passed
* Study A
reg recip dove_f if rapproche==1 & studya==1 & pass85==1
outreg2 using TableD-3.doc, replace ctitle (Model 1)
reg recip dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1 & pass85==1
outreg2 using TableD-3.doc, append ctitle (Model 2)
* Study B
reg recip dove_f if rapproche==1 & studyb==1 & pass85==1
outreg2 using TableD-3.doc, append ctitle (Model 3)
reg recip dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1 & pass85==1
outreg2 using TableD-3.doc, append ctitle (Model 4)

* Table D-5: Main results by Party ID
gen demxdove_f=dem3*dove_f
* Study A
reg recip dove_f if rapproche==1 & pid3==2 & studya==1
outreg2 using TableD-5.doc, replace ctitle (Republicans)
reg recip dove_f if rapproche==1 & pid3==1 & studya==1
outreg2 using TableD-5.doc, append ctitle (Democrats)
reg recip dove_f if rapproche==1 & pid3==3 & studya==1
outreg2 using TableD-5.doc, append ctitle (Independents)
reg recip dove_f dem3 demxdove_f if rapproche==1 & ind3==0 & studya==1
* Study B
reg recip dove_f if rapproche==1 & pid3==2 & studyb==1
outreg2 using TableD-5.doc, append ctitle (Republicans)
reg recip dove_f if rapproche==1 & pid3==1 & studyb==1
outreg2 using TableD-5.doc, append ctitle (Democrats)
reg recip dove_f if rapproche==1 & pid3==3 & studyb==1
outreg2 using TableD-5.doc, append ctitle (Independents)
reg recip dove_f dem3 demxdove_f if rapproche==1 & ind3==0 & studyb==1

* Table D-6: Main results by Trust
gen trustxdove_f=trust*dove_f
* Study A
reg recip dove_f if rapproche==1 & trust==0 & studya==1
outreg2 using TableD-6.doc, replace ctitle (Low trust)
reg recip dove_f if rapproche==1 & trust==1 & studya==1
outreg2 using TableD-6.doc, append ctitle (High trust)
reg recip dove_f trust trustxdove_f if rapproche==1 & studya==1
* Study B
reg recip dove_f if rapproche==1 & trust==0 & studyb==1
outreg2 using TableD-6.doc, append ctitle (Low Trust)
reg recip dove_f if rapproche==1 & trust==1 & studyb==1
outreg2 using TableD-6.doc, append ctitle (High Trust)
reg recip dove_f trust trustxdove_f if rapproche==1 & studyb==1
   
* Table D-7: Main results by Hawkishness
gen hawkishxdove_f=hawkish*dove_f
* Study A
reg recip dove_f if rapproche==1 & hawkish==0 & studya==1
outreg2 using TableD-7.doc, replace ctitle (Low hawkishness)
reg recip dove_f if rapproche==1 & hawkish==1 & studya==1
outreg2 using TableD-7.doc, append ctitle (High hawkishness)
reg recip dove_f hawkish hawkishxdove_f if rapproche==1 & studya==1
* Study B
reg recip dove_f if rapproche==1 & hawkish==0 & studyb==1
outreg2 using TableD-7.doc, append ctitle (Low hawkishness)
reg recip dove_f if rapproche==1 & hawkish==1 & studyb==1
outreg2 using TableD-7.doc, append ctitle (High hawkishness)
reg recip dove_f hawkish hawkishxdove_f if rapproche==1 & studyb==1

* Table F-1: Main results, Elites, Interaction Approach
* Study A
reg recip dove_f dovexmale dovexinc dovexeduc dovexage dovexpolint male highinc higheduc primeage polinthi if rapproche==1 & studya==1
outreg2 using TableF-1.doc, replace ctitle (Model 1)
reg recip dove_f hawk_us dem dovexmale dovexinc dovexeduc dovexage dovexpolint male highinc higheduc primeage polinthi ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableF-1.doc, append ctitle (Model 2)
* Study B
reg recip dove_f dovexmale dovexinc dovexeduc dovexage dovexpolint male highinc higheduc primeage polinthi if rapproche==1 & studyb==1
outreg2 using TableF-1.doc, append ctitle (Model 3)
reg recip dove_f hawk_us dem dovexmale dovexinc dovexeduc dovexage dovexpolint male highinc higheduc primeage polinthi ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableF-1.doc, append ctitle (Model 4)

*Table F-2: Main results, Elites, Subset, no controls, study A
* Study A
reg recip dove_f if rapproche==1 & elite1==1  & studya==1
outreg2 using TableF-2.doc, replace ctitle (Model 1)
reg recip dove_f if rapproche==1 & elite2==1  & studya==1
outreg2 using TableF-2.doc, append ctitle (Model 2)
reg recip dove_f if rapproche==1 & elite3==1  & studya==1
outreg2 using TableF-2.doc, append ctitle (Model 3)
reg recip dove_f if rapproche==1 & elite4==1  & studya==1
outreg2 using TableF-2.doc, append ctitle (Model 4)
reg recip dove_f if rapproche==1 & elite5==1  & studya==1
outreg2 using TableF-2.doc, append ctitle (Model 5)
reg recip dove_f if rapproche==1 & elite6==1  & studya==1
outreg2 using TableF-2.doc, append ctitle (Model 6)

*Table F-3: Main results, Elites, Subset, w/ controls, study A
* Study A
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite1==1 & studya==1
outreg2 using TableF-3.doc, replace ctitle (Model 1)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite2==1 & studya==1
outreg2 using TableF-3.doc, append ctitle (Model 2)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite3==1 & studya==1
outreg2 using TableF-3.doc, append ctitle (Model 3)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite4==1 & studya==1
outreg2 using TableF-3.doc, append ctitle (Model 4)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite5==1 & studya==1
outreg2 using TableF-3.doc, append ctitle (Model 5)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite6==1 & studya==1
outreg2 using TableF-3.doc, append ctitle (Model 6)

*Table F-4: Main results, Elites, Subset, no controls, study B
* Study B
reg recip dove_f if rapproche==1 & elite1==1 & studyb==1
outreg2 using TableF-4.doc, replace ctitle (Model 1)
reg recip dove_f if rapproche==1 & elite2==1 & studyb==1
outreg2 using TableF-4.doc, append ctitle (Model 2)
reg recip dove_f if rapproche==1 & elite3==1 & studyb==1
outreg2 using TableF-4.doc, append ctitle (Model 3)
reg recip dove_f if rapproche==1 & elite4==1 & studyb==1
outreg2 using TableF-4.doc, append ctitle (Model 4)
reg recip dove_f if rapproche==1 & elite5==1 & studyb==1
outreg2 using TableF-4.doc, append ctitle (Model 5)
reg recip dove_f if rapproche==1 & elite6==1 & studyb==1
outreg2 using TableF-4.doc, append ctitle (Model 6)

*Table F-5: Main results, Elites, Subset, w/ controls, study B
* Study B
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite1==1 & studyb==1
outreg2 using TableF-5.doc, replace ctitle (Model 1)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite2==1 & studyb==1
outreg2 using TableF-5.doc, append ctitle (Model 2)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite3==1 & studyb==1
outreg2 using TableF-5.doc, append ctitle (Model 3)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite4==1 & studyb==1
outreg2 using TableF-5.doc, append ctitle (Model 4)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite5==1 & studyb==1
outreg2 using TableF-5.doc, append ctitle (Model 5)
reg recip dove_f ideo7 pid7 white particip relig_impt intl hawkind trust if rapproche==1 & elite6==1 & studyb==1
outreg2 using TableF-5.doc, append ctitle (Model 6)


************************************************************************************************************************************
* M1a (Sincerity 1): Individuals are less likely to believe that a conciliatory gesture is sincere if the foreign leader is a hawk * 
* rather than a dove.                                                                                                              *
************************************************************************************************************************************

* SUPPLEMENTAL (APPENDIX)

* Table C-2: Sincerity Mediator in US-Strong (Study A) 
reg sincere dove_f if rapproche==1 & studya==1 
outreg2 using TableC-2.doc, replace ctitle (Model 1)
reg sincere dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableC-2.doc, append ctitle (Model 2)
reg recip sincere dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableC-2.doc, append ctitle (Model 3)

*Table C-3: Sincerity Mediator in US-Weak (Study B)
reg sincere dove_f if rapproche==1 & studyb==1 
outreg2 using TableC-3.doc, replace ctitle (Model 1)
reg sincere dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1 
outreg2 using TableC-3.doc, append ctitle (Model 2)
reg recip sincere dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1 
outreg2 using TableC-3.doc, append ctitle (Model 3)

* Table C-4: Mediation Analysis for Sincerity
* Study A
local controls "ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust"
medeff (regress sincere dove_f hawk_us dem `controls') (regress recip dove_f sincere hawk_us  dem  `controls') if rapproche==1 & studya==1, treat(dove_f) mediate(sincere) sims(2000) seed(12345)
* Study B
local controls "ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust"
medeff (regress sincere dove_f hawk_us dem `controls') (regress recip dove_f sincere hawk_us  dem  `controls') if rapproche==1 & studyb==1, treat(dove_f) mediate(sincere) sims(2000) seed(12345)


***************************************************************************************************************************************
* H2: As the strength of the foreign leader’s conciliatory gesture increases, the hawk’s disadvantage shrinks.					      *
*                    and																										      *
* H2b (Costly Signaling): Individuals are less likely to disapprove of their leader’s decision to reciprocate a conciliatory gesture  *
* when the foreign gesture is costlier. 																						      *
***************************************************************************************************************************************

* Effect of gesture cost on support for reciprocation (H2b)
tab rapproche if studya==1, sum(recip) 
tab rapproche if studyb==1, sum(recip) 

* FIGURE 2 (in paper)
gen dovex=dovelo
replace dovex=dovelo
reg recip dovex hawkhi dovehi if studya==1 // this is the effect of dove when the gesture is low-cost, Study A
est store s1
est store Strong
replace dovex=dovehi
reg recip dovex hawklo dovelo if studya==1 // this is the effect of dove when the gesture is high-cost, Study A
est store s2
replace dovex=dovelo
reg recip dovex hawkhi dovehi if studyb==1 // this is the effect of dove when the gesture is low-cost, Study B
est store s3
est store Weak
replace dovex=dovehi
reg recip dovex hawklo dovelo if studyb==1 // this is the effect of dove when the gesture is high-cost, Study B
est store s4

coefplot (s1, asequation(When Gesture is Low-Cost)  \ ///
		  s2, asequation(When Gesture is Costly) \ ///
  , keep(dovex) swapnames  xlabel(,labsize(large)) ylabel(,labsize(medium)) mlabel mlabposition(12) mlabgap(*2) mlabformat(%10.1f))  || ///
		 (s3, asequation(When Gesture is Low-Cost) \ ///
		  s4, asequation(When Gesture is Costly) \ ///
  , keep(dovex) swapnames scheme(s1mono) xlabel(,labsize(large)) ylabel(,labsize(medium)) mlabel mlabposition(12) mlabgap(*2) mlabformat(%10.1f)) ||, byopts(compact cols(1))
addplot 1: , subtitle("US-Strong Study") norescaling
addplot 2: , subtitle("US-Weak Study") norescaling

* SUPPLEMENTAL (APPENDIX)

* Table C-5: Gesture Strength
* Study A
reg recip dovelo dovehi hawklo hawkhi if studya==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-5.doc, replace ctitle (Model 1)
reg recip dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studya==1 ,  noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-5.doc, append ctitle (Model 2)
* Study B
reg recip dovelo dovehi hawklo hawkhi if studyb==1 , noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-5.doc, append ctitle (Model 3)
reg recip dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studyb==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-5.doc, append ctitle (Model 4)

*Table D-2: Gesture Strength - Number of Bases
* Study A
reg numberbases dovelo dovehi hawklo hawkhi if studya==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-2.doc, replace ctitle (Model 1)
reg numberbases dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studya==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-2.doc, append ctitle (Model 2)
* Study B
reg numberbases dovelo dovehi hawklo hawkhi if studyb==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-2.doc, append ctitle (Model 3)
reg numberbases dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studyb==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-2.doc, append ctitle (Model 4)

* Table D-4: Gesture Strength - 85% passed
* Study A
reg recip dovelo dovehi hawklo hawkhi if pass85==1 & studya==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-4.doc, replace ctitle (Model 1)
reg recip dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if pass85==1 & studya==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-4.doc, append ctitle (Model 2)
* Study B
reg recip dovelo dovehi hawklo hawkhi if pass85==1 & studyb==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-4.doc, append ctitle (Model 3)
reg recip dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if pass85==1 & studyb==1, noconstant 
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableD-4.doc, append ctitle (Model 4)



*******************************************************************************************************************************************
* H2a (Sincerity 2): As the strength of the foreign leader’s conciliatory gesture increases, the gap in individuals’ perceptions of the   * 
* sincerity of conciliatory gestures by foreign hawks versus by foreign doves shrinks                                                     *
*******************************************************************************************************************************************

*Table C-6: Difference in Sincerity Perceptions Across Weak and Strong Gesture
* Study A
reg sincere dovelo dovehi hawklo hawkhi if studya==1, noconstant
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-6.doc, replace ctitle (Model 1)
reg sincere dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studya==1, noconstant
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-6.doc, append ctitle (Model 2)
* Study B
reg sincere dovelo dovehi hawklo hawkhi if studyb==1, noconstant
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-6.doc, append ctitle (Model 3)
reg sincere dovelo dovehi hawklo hawkhi hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studyb==1, noconstant
lincom dovehi-hawkhi 
lincom dovelo-hawklo
lincom dovehi-hawkhi-dovelo+hawklo
outreg2 using TableC-6.doc, append ctitle (Model 4)


*Table C-7: Effect of Perception of Sincerity on Support for Reciprocation, Costly Gesture
* Study A
reg recip sincere dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studya==1
outreg2 using TableC-7.doc, replace ctitle (Model 1)
* Study B
reg recip sincere dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studyb==1
outreg2 using TableC-7.doc, append ctitle (Model 2)


*********************************************************
***** Appendix E: Additional Hypotheses in PAP **********
*********************************************************


** Exploitation Mechanism

* Table E-A-1: Exploit, Study A 
* Study A
reg exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableE-A-1.doc, replace ctitle (Model 1)
reg recip exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableE-A-1.doc, append ctitle (Model 2)
reg exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studya==1
outreg2 using TableE-A-1.doc, append ctitle (Model 3)
reg recip exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studya==1
outreg2 using TableE-A-1.doc, append ctitle (Model 4)

* Table E-A-2: Exploit Study B 
* Study B
reg exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableE-A-2.doc, replace ctitle (Model 1)
reg recip exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableE-A-2.doc, append ctitle (Model 2)
reg exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studyb==1
outreg2 using TableE-A-2.doc, append ctitle (Model 3)
reg recip exploit dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studyb==1
outreg2 using TableE-A-2.doc, append ctitle (Model 4)

* Table E-A-3: Trick, Study A
reg trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableE-A-3.doc, replace ctitle (Model 1)
reg recip trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableE-A-3.doc, append ctitle (Model 2)
reg trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studya==1
outreg2 using TableE-A-3.doc, append ctitle (Model 3)
reg recip trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studya==1
outreg2 using TableE-A-3.doc, append ctitle (Model 4)

* Table E-A-4: Trick, Study B
reg trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableE-A-4.doc, replace ctitle (Model 1)
reg recip trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableE-A-4.doc, append ctitle (Model 2)
reg trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studyb==1
outreg2 using TableE-A-4.doc, append ctitle (Model 3)                                                                                                     
reg recip trick dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studyb==1
outreg2 using TableE-A-4.doc, append ctitle (Model 4)


** Fear Mechanism

* Table E-B-1: Fear Mechanism, Study A
reg fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableE-B-1.doc, replace ctitle (Model 1)
reg recip fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studya==1
outreg2 using TableE-B-1.doc, append ctitle (Model 2)
reg fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studya==1
outreg2 using TableE-B-1.doc, append ctitle (Model 3)
reg recip fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studya==1
outreg2 using TableE-B-1.doc, append ctitle (Model 4)

*Table E-B-2: Fear Mechanism, Study B
reg fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableE-B-2.doc, replace ctitle (Model 1)
reg recip fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==1 & studyb==1
outreg2 using TableE-B-2.doc, append ctitle (Model 2)
reg fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studyb==1
outreg2 using TableE-B-2.doc, append ctitle (Model 3)
reg recip fear dove_f hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if rapproche==2 & studyb==1
outreg2 using TableE-B-2.doc, append ctitle (Model 4)


** Threat Moderator

*Table E-C-1: Threat
gen dove_en=0
replace dove_en=1 if dove_f==1 & enemy2==1
gen hawk_en=0
replace hawk_en=1 if dove_f==0 & enemy2==1
gen dove_fr=0
replace dove_fr=1 if dove_f==1 & enemy2==0
gen hawk_fr=0
replace hawk_fr=1 if dove_f==0 & enemy2==0
* Study A
reg recip hawk_fr dove_fr hawk_en dove_en hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studya==1, noconstant
lincom dove_en-hawk_en 
lincom dove_fr- hawk_en
lincom dove_en-hawk_en - dove_fr + hawk_fr
outreg2 using TableE-C-1.doc, replace ctitle (Model 1)
* Study B
reg recip hawk_fr dove_fr hawk_en dove_en hawk_us dem ideo7 pid7 male age white educ polint particip relig_impt intl hawkind trust if studyb==1, noconstant
lincom dove_en-hawk_en 
lincom dove_fr- hawk_en
lincom dove_en-hawk_en - dove_fr + hawk_fr
outreg2 using TableE-C-1.doc, append ctitle (Model 2)

log close


